counting.nb

This notebook accompanies the paper M. Levitin, I. Polterovich, D. A. Sher, *Pólya's conjecture for the disk: a computer-assisted proof*, available at https://arxiv.org/abs/2203.07696, see also https://michaellevitin.net/polya.html

For simplicity, this notebook differs slightly from the one used in the actual calculations described in the paper, by removing some non-essential embellishments. As a result, the output may also be slightly different from that available for download at https://michaellevitin.net/polya.html

This script reads files smallqminus.csv, smallqplus.csv, capitalqminus.csv, capitalqplus.csv, and capitalqr.csv from https://michaellevitin.net/polya.html

---

Notebook initialisation

```
SetDirectory[NotebookDirectory[]];
<< MaTeX`
texStyle = {};
SetOptions[MaTeX,
  "BasePreamble" → {"\\usepackage{amsmath}", "\\usepackage{xcolor}",
    "\\usepackage{fourier}", "\\usepackage{ebgaramond}"}, FontSize → 11];
```

Loading the coefficients of $\underline{h}_{\mathbb{Q}}$ and $\overline{h}_{\mathbb{Q}}$ and converting them into fractions (compare with Table 4 in the paper)

In[6]:=
```
qdataminus = Import["https://michaellevitin.net/PolyaData/smallqminus.csv"];
qdataplus = Import["https://michaellevitin.net/PolyaData/smallqplus.csv"];
Qdataminus = Import["https://michaellevitin.net/PolyaData/capitalqminus.csv"];
Qdataplus = Import["https://michaellevitin.net/PolyaData/capitalqplus.csv"];
QdataR = Import["https://michaellevitin.net/PolyaData/capitalqr.csv"];
```

In what follows qminus, qplus, Qminus, Qplus, and QR are the indexed quantities q_-, q_+, Q_-, Q_+ and Q_R from the paper, with the second index shifted by one, so that qminus[[i]][[j]]=$q_{-,i,j-1}$, etc.

In[11]:=
```
qminus = Table[qdataminus[[i, j]] / qdataminus[[i, j + 1]], {i, 14}, {j, 1, 11, 2}];
qminus // TableForm
qplus = Table[qdataplus[[i, j]] / qdataplus[[i, j + 1]], {i, 14}, {j, 1, 11, 2}];
qplus // TableForm
```

Out[12]//TableForm=

1475/5022	−2569/5307	523/3282	9/3380	97/7232	8/6631
378/1531	−2127/4705	1570/9753	9/1093	37/2522	18/4573
86/423	−2084/4967	490/2981	25/1711	57/3251	31/3978
290/1779	−626/1621	429/2525	119/5268	41/1791	33/2311
155/1229	−1444/4109	1283/7199	119/3550	25/762	115/4272
237/2554	−342/1087	400/2099	28/559	80/1527	417/7490
115/1636	−794/2785	872/4277	101/1449	174/2135	219/2057
254/4491	−649/2457	530/2457	347/3892	203/1751	272/1541
212/4829	−1015/4197	748/3237	467/3967	604/3441	1284/4039
36/1111	−687/3155	205/814	101/620	1529/5279	2077/3227
181/8165	−1564/8181	777/2759	1672/6895	1412/2599	15 016/9655
51/3821	−411/2555	741/2254	961/2349	2877/2279	7097/1402
8/1293	−431/3474	524/1251	4199/4694	18 111/3989	130 957/4318
7/5895	−197/2762	3433/4793	29 881/6338	164 653/2318	4 383 291/3083

Out[14]//TableForm=

1475/5022	−1733/3580	523/3282	5/1878	97/7232	4/3317
378/1531	−1363/3015	1570/9753	20/2429	37/2522	13/3303
86/423	−819/1952	490/2981	43/2943	57/3251	79/10 138
290/1779	−531/1375	429/2525	63/2789	41/1791	31/2171
155/1229	−1172/3335	1283/7199	155/4624	25/762	101/3752
237/2554	−499/1586	400/2099	225/4492	80/1527	183/3287
115/1636	−1192/4181	872/4277	251/3601	174/2135	275/2583
254/4491	−705/2669	530/2457	865/9702	203/1751	275/1558
212/4829	−1104/4565	748/3237	469/3984	604/3441	872/2743
36/1111	−871/4000	205/814	642/3941	1529/5279	475/738
181/8165	−2214/11 581	777/2759	1898/7827	1412/2599	12 680/8153
51/3821	−351/2182	741/2254	988/2415	2877/2279	13 460/2659
8/1293	−481/3877	524/1251	14 777/16 519	18 111/3989	326 331/10 760
7/5895	−99/1388	3433/4793	11 546/2449	164 653/2318	7 529 649/5296

In[15]:=
```
Qminus = Table[Qdataminus[[i, j]] / Qdataminus[[i, j + 1]], {i, 13}, {j, 1, 11, 2}];
Qplus = Table[Qdataplus[[i, j]] / Qdataplus[[i, j + 1]], {i, 13}, {j, 1, 11, 2}];
QR = Table[QdataR[[i, j]] / QdataR[[i, j + 1]], {i, 10}, {j, 1, 3, 2}];
Qminus // TableForm
Qplus // TableForm
QR // TableForm
```

Out[18]//TableForm=

$\frac{1097}{3735}$	$-\frac{1733}{3580}$	$\frac{316}{1983}$	$\frac{5}{1878}$	$\frac{56}{4175}$	0
$\frac{776}{3143}$	$-\frac{1363}{3015}$	$\frac{844}{5243}$	$\frac{20}{2429}$	$\frac{25}{1704}$	$\frac{4}{1607}$
$\frac{995}{4894}$	$-\frac{819}{1952}$	$\frac{514}{3127}$	$\frac{43}{2943}$	$\frac{59}{3365}$	$\frac{13}{2299}$
$\frac{647}{3969}$	$-\frac{531}{1375}$	$\frac{569}{3349}$	$\frac{63}{2789}$	$\frac{85}{3713}$	$\frac{13}{1231}$
$\frac{493}{3909}$	$-\frac{1172}{3335}$	$\frac{977}{5482}$	$\frac{155}{4624}$	$\frac{98}{2987}$	$\frac{52}{2673}$
$\frac{76}{819}$	$-\frac{499}{1586}$	$\frac{796}{4177}$	$\frac{225}{4492}$	$\frac{126}{2405}$	$\frac{259}{6795}$
$\frac{177}{2518}$	$-\frac{1192}{4181}$	$\frac{1061}{5204}$	$\frac{251}{3601}$	$\frac{137}{1681}$	$\frac{306}{3613}$
$\frac{163}{2882}$	$-\frac{705}{2669}$	$\frac{1222}{5665}$	$\frac{865}{9702}$	$\frac{430}{3709}$	$\frac{579}{4261}$
$\frac{410}{9339}$	$-\frac{1104}{4565}$	$\frac{632}{2735}$	$\frac{469}{3984}$	$\frac{835}{4757}$	$\frac{469}{2004}$
$\frac{79}{2438}$	$-\frac{871}{4000}$	$\frac{854}{3391}$	$\frac{642}{3941}$	$\frac{601}{2075}$	$\frac{1401}{3152}$
$\frac{73}{3293}$	$-\frac{2214}{11581}$	$\frac{541}{1921}$	$\frac{1898}{7827}$	$\frac{615}{1132}$	$\frac{1716}{1763}$
$\frac{37}{2772}$	$-\frac{351}{2182}$	$\frac{1267}{3854}$	$\frac{988}{2415}$	$\frac{611}{484}$	$\frac{3352}{1255}$
$\frac{18}{2909}$	$-\frac{481}{3877}$	$\frac{1794}{4283}$	$\frac{14777}{16519}$	$\frac{7278}{1603}$	$\frac{37396}{3379}$

Out[19]//TableForm=

$\frac{1097}{3735}$	$-\frac{2569}{5307}$	$\frac{316}{1983}$	$\frac{9}{3380}$	$\frac{56}{4175}$	$\frac{10}{4017}$
$\frac{776}{3143}$	$-\frac{2127}{4705}$	$\frac{844}{5243}$	$\frac{9}{1093}$	$\frac{25}{1704}$	$\frac{6}{1061}$
$\frac{995}{4894}$	$-\frac{2084}{4967}$	$\frac{514}{3127}$	$\frac{25}{1711}$	$\frac{59}{3365}$	$\frac{45}{4261}$
$\frac{647}{3969}$	$-\frac{626}{1621}$	$\frac{569}{3349}$	$\frac{119}{5268}$	$\frac{85}{3713}$	$\frac{72}{3701}$
$\frac{493}{3909}$	$-\frac{1444}{4109}$	$\frac{977}{5482}$	$\frac{119}{3550}$	$\frac{98}{2987}$	$\frac{319}{8369}$
$\frac{76}{819}$	$-\frac{342}{1087}$	$\frac{796}{4177}$	$\frac{28}{559}$	$\frac{126}{2405}$	$\frac{140}{1653}$
$\frac{177}{2518}$	$-\frac{794}{2785}$	$\frac{1061}{5204}$	$\frac{101}{1449}$	$\frac{137}{1681}$	$\frac{760}{5593}$
$\frac{163}{2882}$	$-\frac{649}{2457}$	$\frac{1222}{5665}$	$\frac{347}{3892}$	$\frac{430}{3709}$	$\frac{491}{2098}$
$\frac{410}{9339}$	$-\frac{1015}{4197}$	$\frac{632}{2735}$	$\frac{467}{3967}$	$\frac{835}{4757}$	$\frac{1381}{3107}$
$\frac{79}{2438}$	$-\frac{687}{3155}$	$\frac{854}{3391}$	$\frac{101}{620}$	$\frac{601}{2075}$	$\frac{1570}{1613}$
$\frac{73}{3293}$	$-\frac{1564}{8181}$	$\frac{541}{1921}$	$\frac{1672}{6895}$	$\frac{615}{1132}$	$\frac{6493}{2431}$
$\frac{37}{2772}$	$-\frac{411}{2555}$	$\frac{1267}{3854}$	$\frac{961}{2349}$	$\frac{611}{484}$	$\frac{48596}{4391}$
$\frac{18}{2909}$	$-\frac{431}{3474}$	$\frac{1794}{4283}$	$\frac{4199}{4694}$	$\frac{7278}{1603}$	$\frac{1005239}{8009}$

Out[20]//TableForm=

$\frac{15}{4459}$	$-\frac{528}{5369}$
$\frac{4}{1393}$	$-\frac{158}{1699}$
$\frac{8}{3325}$	$-\frac{122}{1397}$
$\frac{11}{5587}$	$-\frac{286}{3519}$
$\frac{4}{2561}$	$-\frac{155}{2074}$
$\frac{3}{2525}$	$-\frac{274}{4055}$
$\frac{2}{2353}$	$-\frac{115}{1931}$
$\frac{2}{3623}$	$-\frac{100}{1989}$
$\frac{1}{3327}$	$-\frac{77}{1984}$
$\frac{1}{9390}$	$-\frac{52}{2451}$

Creating partitions of [0, 1] and lists of functions composing $\underline{h}_{\mathbb{Q}}$ and $\overline{h}_{\mathbb{Q}}$

In[21]:=
```
xs = Join[Table[(i - 1) / 10, {i, 1, 7}], Table[6 / 10 + (i - 7) / 20, {i, 8, 15}]]
 (* points partitioning [0,1] into 14 original intervals E_i *)
```

Out[21]=

$$\left\{0, \frac{1}{10}, \frac{1}{5}, \frac{3}{10}, \frac{2}{5}, \frac{1}{2}, \frac{3}{5}, \frac{13}{20}, \frac{7}{10}, \frac{3}{4}, \frac{4}{5}, \frac{17}{20}, \frac{9}{10}, \frac{19}{20}, 1\right\}$$

In[22]:=
```
xcs = Table[1 / 2 (xs[[i]] + xs[[i + 1]]), {i, 14}] (* centres of the intervals,
 partitioning them further into subintervals E_∓ *)
```

Out[22]=

$$\left\{\frac{1}{20}, \frac{3}{20}, \frac{1}{4}, \frac{7}{20}, \frac{9}{20}, \frac{11}{20}, \frac{5}{8}, \frac{27}{40}, \frac{29}{40}, \frac{31}{40}, \frac{33}{40}, \frac{7}{8}, \frac{37}{40}, \frac{39}{40}\right\}$$

In[23]:=
```
xRs = Table[95 / 100 + (k - 1) / 200, {k, 1, 11}]
 (* subpartition of the right-most interval E14=[95/100,1] *)
```

Out[23]=

$$\left\{\frac{19}{20}, \frac{191}{200}, \frac{24}{25}, \frac{193}{200}, \frac{97}{100}, \frac{39}{40}, \frac{49}{50}, \frac{197}{200}, \frac{99}{100}, \frac{199}{200}, 1\right\}$$

In[24]:=
```
xQs = Union[xs, xcs, xRs] // Sort
(* all subinterval partition points for h̄_Q; 36 subintervals in total *)
xQs // Length
```

Out[24]=

$$\left\{0, \frac{1}{20}, \frac{1}{10}, \frac{3}{20}, \frac{1}{5}, \frac{1}{4}, \frac{3}{10}, \frac{7}{20}, \frac{2}{5}, \frac{9}{20}, \frac{1}{2}, \frac{11}{20}, \frac{3}{5}, \frac{5}{8}, \frac{13}{20}, \frac{27}{40}, \frac{7}{10}, \frac{29}{40}, \frac{3}{4}, \frac{31}{40}, \frac{4}{5}, \frac{33}{40}, \frac{17}{20}, \frac{7}{8}, \frac{9}{10}, \frac{37}{40}, \frac{19}{20}, \frac{191}{200}, \frac{24}{25}, \frac{193}{200}, \frac{97}{100}, \frac{39}{40}, \frac{49}{50}, \frac{197}{200}, \frac{99}{100}, \frac{199}{200}, 1\right\}$$

Out[25]=

37

In[26]:=
```
xqs = Union[xs, xcs] // Sort
(* all subinterval partition points for hQ; 28 subintervals in total *)
xqs // Length
```

Out[26]=
$$\left\{0, \frac{1}{20}, \frac{1}{10}, \frac{3}{20}, \frac{1}{5}, \frac{1}{4}, \frac{3}{10}, \frac{7}{20}, \frac{2}{5}, \frac{9}{20}, \frac{1}{2}, \frac{11}{20}, \frac{3}{5}, \frac{5}{8}, \frac{13}{20}, \frac{27}{40}, \frac{7}{10}, \frac{29}{40}, \frac{3}{4}, \frac{31}{40}, \frac{4}{5}, \frac{33}{40}, \frac{17}{20}, \frac{7}{8}, \frac{9}{10}, \frac{37}{40}, \frac{19}{20}, \frac{39}{40}, 1\right\}$$

Out[27]=
```
29
```

The following are the lists of expressions (also defined as functions later) for $\underline{h}_{\mathbb{Q}}$ and $\overline{h}_{\mathbb{Q}}$ on subintervals. The list for $\underline{h}_{\mathbb{Q}}$ is $\{\underline{h}_{\mathbb{Q},1,-}, \underline{h}_{\mathbb{Q},1,+}, \underline{h}_{\mathbb{Q},2,-}, \underline{h}_{\mathbb{Q},2,+}, \ldots, \underline{h}_{\mathbb{Q},14,-}, \underline{h}_{\mathbb{Q},14,+}\}$, and the list for $\overline{h}_{\mathbb{Q}}$ is $\{\overline{h}_{\mathbb{Q},1,-}, \overline{h}_{\mathbb{Q},1,+}, \ldots, \overline{h}_{\mathbb{Q},13,-}, \overline{h}_{\mathbb{Q},13,+}, \overline{h}_{\mathbb{Q},R,1}, \ldots, \overline{h}_{\mathbb{Q},R,10}\}$, where

$$\underline{h}_{\mathbb{Q},i,\pm} = q_{i,0,\pm} + \sum_{j=1}^{5} q_{i,j,\pm}(x - x_{c,i})^j, \qquad \overline{h}_{\mathbb{Q},i,\pm} = Q_{i,0,\pm} + \sum_{j=1}^{5} Q_{i,j,\pm}(x - x_{c,i})^j, \qquad \overline{h}_{\mathbb{Q},R,k} = Q_{R,k,0} + Q_{R,k,1}(x - x_{R,k})$$

In[28]:=
```
hDown =
  Flatten[Table[{qminus[[i]][[1]] + Sum[qminus[[i]][[j]] (x - xcs[[i]]) ^ (j - 1), {j, 2, 6}],
     qplus[[i]][[1]] + Sum[qplus[[i]][[j]] (x - xcs[[i]]) ^ (j - 1), {j, 2, 6}]}, {i, 1, 14}]];
hDownfns = Table[Function[x, hDown[[i]] // Evaluate], {i, 1, 28}];
hUp = Join[Flatten[Table[
     {Qminus[[i]][[1]] + Sum[Qminus[[i]][[j]] (x - xcs[[i]]) ^ (j - 1), {j, 2, 6}], Qplus[[i]][[1]] +
       Sum[Qplus[[i]][[j]] (x - xcs[[i]]) ^ (j - 1), {j, 2, 6}]}, {i, 1, 13}]],
   Table[QR[[k]][[1]] + QR[[k]][[2]] (x - xRs[[k]]), {k, 1, 10}]
  ];
hUpfns = Table[Function[x, hUp[[i]] // Evaluate], {i, 1, 36}];
```

Checking the inequalities $\underline{h}_{\mathbb{Q}}(x) \le h(x) \le \overline{h}_{\mathbb{Q}}(x)$

This section is for illustrative purposes only. We demonstrate (numerically) the validity of the inequalities above, cf. Figure 9 in the paper. We omit some embellishments made in the paper in order to improve the graphics presentation.

In[50]:=
```
h[x_] := 1 / Pi (Sqrt[1 - x^2] - x ArcCos[x]);
PiecewisehDown = Append[Table[{hDown[[i]], xqs[[i]] ≤ x < xqs[[i + 1]]}, {i, 1, 27}],
   {hDown[[28]], xqs[[28]] ≤ x ≤ xqs[[29]]}];
PiecewisehUp = Append[Table[{hUp[[i]], xQs[[i]] ≤ x < xQs[[i + 1]]}, {i, 1, 35}],
   {hUp[[36]], xQs[[36]] ≤ x ≤ xQs[[37]]}];
xticks1 = {{0.3, 0.6, 0.9}, MaTeX@{"0.3", "0.6", "0.9"}} // Transpose;
xticks2 = {{0.9, 0.95, 1}, MaTeX@{"0.9", "0.95", "1"}} // Transpose;
yticks1 = {{1, 2, 3} 10^(-7),
    MaTeX@{"10^{-7}", "2\\cdot 10^{-7}", "3\\cdot 10^{-7}"}} // Transpose;
yticks2 = {{3, 6, 9} 10^(-6), MaTeX@
     {"3\\cdot 10^{-6}", "6\\cdot 10^{-6}", "9\\cdot 10^{-6}"}} // Transpose;
yticks3 = {{1, 2} 10^(-6), MaTeX@{"10^{-6}", "2\\cdot 10^{-6}"}} // Transpose;
xlabel = MaTeX["x"];
ylabel1 = MaTeX["\\overline{h}_{\\mathbb{Q}}(x)-h(x)"];
ylabel2 = MaTeX["\\underline{h}_{\\mathbb{Q}}(x)-h(x)"];
GraphicsGrid[
 {{Plot[Piecewise[PiecewisehUp] - h[x], {x, 0, 0.9}, PlotRange → Automatic,
    PlotStyle → Blue, AxesOrigin → {0, 0}, Ticks → {xticks1, yticks1},
    AxesLabel → {xlabel, ylabel1}],
   Plot[Piecewise[PiecewisehUp] - h[x], {x, 0.9, 1},
    PlotRange → Automatic, PlotStyle → Blue, AxesOrigin → {0.895, 0},
    Ticks → {xticks2, yticks2}, AxesLabel → {xlabel, ylabel1}]},
  {Plot[h[x] - Piecewise[PiecewisehDown], {x, 0, 0.9}, PlotRange → Automatic,
    PlotStyle → Red, AxesOrigin → {0, 0}, Ticks → {xticks1, yticks1},
    AxesLabel → {xlabel, ylabel2}], Plot[h[x] - Piecewise[PiecewisehDown],
    {x, 0.9, 1}, PlotRange → Automatic, PlotStyle → Red, AxesOrigin → {0.895, 0},
    Ticks → {xticks2, yticks3}, AxesLabel → {xlabel, ylabel2}]}}]
```

Out[61]=



Counting the points and setting up the iteration processes

The next two functions compute $\overline{\mathcal{P}}^{D}(\lambda)$ and $\underline{\mathcal{P}}^{N}(\lambda)$ directly

In[62]:=
```
PCountD[λ_] := Module[{m, i, mP, P, Nf},
  Nf = 36;
  mP = Floor[xQs λ]; mP[[1]] = 0;
  P = Sum[Floor[λ hUpfns[[i]][m / λ] + 1 / 4], {i, 1, Nf}, {m, mP[[i]] + 1, mP[[i + 1]]}];
  P = 2 P + Floor[λ hUpfns[[1]][0] + 1 / 4]
 ]
PCountN[λ_] := Module[{m, i, mP, P, Nf},
  Nf = 28;
  mP = Floor[xqs λ]; mP[[1]] = 0;
  P = Sum[Floor[λ hDownfns[[i]][m / λ] + 3 / 4], {i, 1, Nf}, {m, mP[[i]] + 1, mP[[i + 1]]}];
  2 P + Floor[λ hDownfns[[1]][0] + 3 / 4]
 ]
```

Example:

In[64]:=
```
Λ1 = 84 123;
pd1 = PCountD[Λ1]
Λ1^2 / 4 - pd1
pn1 = PCountN[Λ1]
pn1 - Λ1^2 / 4
```

Out[65]=
$$1\,769\,133\,931$$

Out[66]=
$$\frac{143\,405}{4}$$

Out[67]=
$$1\,769\,207\,179$$

Out[68]=
$$\frac{149\,587}{4}$$

The following auxiliary function Fori[α0, d] realises Forišek's algorithm (see the citation in the paper) of finding the smallest possible denominator rational approximation to a rational number α0>0 within the interval (α0-d, α0+d). Its typical uses are Fori[α0+ϵ, δ] with small rational values $0<\delta<\epsilon$ to give a rational approximation of α0 from above with a small denominator, and Fori[α0-ϵ, δ] to give a rational approximation of α0 from below with a small denominator. Using Fori is not, strictly speaking, necessary, but ensures that the denominators of the fractions appearing in the iteration processes are kept relatively small, at the cost of slightly slowing the processes.

In[69]:=
```
Fori[α0_, d_] := Module[{α, denom, αnum, dnum, xnum, xdenom, x, pa, qa, pb,
    qb, aa, bb, cc, k, knum, kdenom, nsteps, newpa, newpb, newqa, newqb},
  α = α0 - Floor[α0];
```

```
 If[α == 0, Return[α0]];
 denom = LCM[Denominator[α], Denominator[d] ];
 αnum = α denom; dnum = d denom;
 pa = 0; qa = 1; pb = 1; qb = 1; nsteps = 0;
 While[nsteps ≤ 1000,
  xnum = denom pb - αnum qb;
  xdenom = -denom pa + αnum qa;
  x = Ceiling[xnum / xdenom];
  aa = (α - d < (pb + x pa) / (qb + x qa) < α + d);
  bb = (α - d < (pb + (x - 1) pa) / (qb + (x - 1) qa) < α + d);
  If[(aa || bb),
   knum = denom pb - (αnum + dnum) qb;
   kdenom = (αnum + dnum) qa - denom pa;
   k = Ceiling[knum / kdenom];
   cc = (pb + k pa) / (qb + k qa); Break[]
  ];
  newpa = pb + (x - 1) pa;
  newqa = qb + (x - 1) qa;
  newpb = pb + x pa;
  newqb = qb + x qa;
  pa = newpa;
  pb = newpb;
  qa = newqa;
  qb = newqb;
  (* *************** *)
  xnum = -denom pb + αnum qb;
  xdenom = denom pa - αnum qa;
  x = Ceiling[xnum / xdenom];
  aa = (α - d < (pb + x pa) / (qb + x qa) < α + d);
  bb = (α - d < (pb + (x - 1) pa) / (qb + (x - 1) qa) < α + d);
  If[(aa || bb),
   knum = -denom pb + (αnum - dnum) qb;
   kdenom = -(αnum - dnum) qa + denom pa;
   k = Ceiling[knum / kdenom];
   cc = (pb + k pa) / (qb + k qa); Break[]
  ];
  newpa = pb + (x - 1) pa;
  newqa = qb + (x - 1) qa;
  newpb = pb + x pa;
  newqb = qb + x qa;
  pa = newpa;
  pb = newpb;
  qa = newqa;
  qb = newqb;
  nsteps++;
 ];
 cc + Floor[α0]
];
```

Example of the use of Fori:

In[70]:=
```
α0 = 14 153 214 853 / 168 246
αAbove = Fori[α0 + 1 / 100, 1 / 101]
αBelow = Fori[α0 - 1 / 100, 1 / 101]
αBelow < α0 < αAbove
```

Out[70]=
$$\frac{14\,153\,214\,853}{168\,246}$$

Out[71]=
$$\frac{504\,733}{6}$$

Out[72]=
$$\frac{588\,855}{7}$$

Out[73]=
```
True
```

The following two functions IterD and IterN each perform one step of the iteration process for the Dirichlet and Neumann point counts, respectively, at a point λ, see Figure 7. The output in the Dirichlet case is $\{\lambda,\ \overline{\mathcal{P}}^D(\lambda), e^D(\lambda),\ \text{shift}^D(\lambda),\ \lambda_{\text{new}}\}$, where

$e^D(\lambda) = \lambda^2/4 - \overline{\mathcal{P}}^D(\lambda)$, $\quad \text{shift}^D(\lambda) = 2\, e^D(\lambda)/\lambda$, $\quad \lambda_{\text{new}} = \text{Fori}[\lambda - \text{shift}^D(\lambda) + 1/100, 1/101]$.

The output in the Neumann case is $\{\lambda,\ \underline{\mathcal{P}}^N(\lambda), e^N(\lambda),\ \text{shift}^N(\lambda),\ \lambda_{\text{new}}\}$, where

$e^N(\lambda) = \underline{\mathcal{P}}^N(\lambda) - \lambda^2/4$, $\quad \text{shift}^N(\lambda) = 2\, e^N(\lambda)/(\lambda + 2\, e^N(\lambda)/\lambda)$, $\quad \lambda_{\text{new}} = \text{Fori}[\lambda + \text{shift}^N(\lambda) - 1/100,\ 1/101]$.

In[74]:=
```
IterD[λ_] := Module[{err, λnew, shift, pD},
   pD = PCountD[λ];
   err = λ^2 / 4 - pD;
   shift = 2 err / λ;
   λnew = Fori[λ - shift + 1 / 100, 1 / 101];
   {λ, pD, err, shift, λnew}
  ];
IterN[λ_] := Module[{err, λnew, shift, pN},
   pN = PCountN[λ];
   err = pN - λ^2 / 4;
   shift = 2 err / (λ + 2 err / λ);
   λnew = Fori[λ + shift - 1 / 100, 1 / 101];
   {λ, pN, err, shift, λnew}
  ];
```

Example:

In[76]:=
```
IterD[Λ1]
```

Out[76]=

$$\left\{84\,123,\ 1\,769\,133\,931,\ \frac{143\,405}{4},\ \frac{143\,405}{168\,246},\ \frac{504\,733}{6}\right\}$$

In[77]:=
```
IterN[5 / 2]
```

Out[77]=

$$\left\{\frac{5}{2},\ 3,\ \frac{23}{16},\ \frac{115}{146},\ \frac{23}{7}\right\}$$

Actual calculations

The following functions (shown here without commands used for output into external files) have been used to perform the actual calculations. The parameter freq in both functions stands for the frequency of screen output, thus taking freq=1 will output on each step, and freq=100 on every hundredth step. The parameter id is a string used to identify the output from a particular calculation (useful in parallelised processes). For ease of observation only, some output is converted into floating point numbers.

In[81]:=
```
DoIterationsD[ΛStart_, ΛMin_, id_, freq_] :=
  Module[{nI, nIfinal, times0, err, Λnew, Λ, pD, shift, time},
   times0 = TimeUsed[];
   Λ = ΛStart;
   Do[
    {Λ, pD, err, shift, Λnew} = IterD[Λ];
    time = TimeUsed[] - times0;
    If[Mod[nI, freq] == 0, Print[id <> ": ", nI,
      " iterations    ", "      λ=", N[Λnew, 7], "      time=", time]];
    Λ = Λnew;
    If[shift ≤ 0, Return[id <> ": FAILS at λ=" <> ToString[Λ]]];
    If[Λ < ΛMin, nIfinal = nI; Break[]];
    ,
    {nI, 1, ∞}
   ];
   id <> ": FINISHED!!!" <> " in time " <> ToString[time] <>
    " sec  after " <> ToString[nIfinal] <> "  iterations"
  ];
DoIterationsN[ΛStart_, ΛMax_, id_, freq_] :=
  Module[{nI, nIfinal, times0, err, Λnew, Λ, pN, shift, time},
   times0 = TimeUsed[];
   Λ = ΛStart;
   Do[
    {Λ, pN, err, shift, Λnew} = IterN[Λ];
    time = TimeUsed[] - times0;
    If[Mod[nI, freq] == 0, Print[id <> ": ", nI,
      " iterations    ", "      λ=", N[Λnew, 7], "      time=", time]];
    Λ = Λnew;
    If[shift ≤ 0, Return[id <> ": FAILS at λ=" <> ToString[Λ]]];
    If[Λ > ΛMax, nIfinal = nI; Break[]];
    ,
    {nI, 1, ∞}
   ];
   id <> ": FINISHED!!!" <> " in time " <> ToString[time] <>
    " sec  after " <> ToString[nIfinal] <> "  iterations"
  ];
```

Examples:

In[83]:=
```
DoIterationsD[84123, 84100, "Dirichlet Test", 5]
```

```
Dirichlet Test: 5 iterations          λ=84118.83      time=29.4504
Dirichlet Test: 10 iterations          λ=84114.63      time=63.2083
Dirichlet Test: 15 iterations          λ=84110.43      time=97.6523
Dirichlet Test: 20 iterations          λ=84106.25      time=129.679
Dirichlet Test: 25 iterations          λ=84102.06      time=165.681
```

Out[83]=
```
Dirichlet Test: FINISHED!!! in time 186.137 sec  after 28  iterations
```

In[84]:=
```
DoIterationsN[84 100, 84 123, "Neumann Test", 5]
```

```
Neumann Test: 5 iterations          λ=84 104.38      time=29.7945
Neumann Test: 10 iterations          λ=84 108.77      time=63.893
Neumann Test: 15 iterations          λ=84 113.17      time=100.458
Neumann Test: 20 iterations          λ=84 117.57      time=134.728
Neumann Test: 25 iterations          λ=84 121.94      time=170.362
```

Out[84]=
```
Neumann Test: FINISHED!!! in time 183.872 sec  after 27  iterations
```

The full computer-assisted proof can be verified, for example, by executing the following command (WARNING: may take a very long time, especially on slower computers; ignore the indication of remaining time)

```
Λ1 = 84 123; Λ0 = 5 / 2;
Parallelize[{
  DoIterationsD[80 000, 70 000, "Dirichlet 80K-70K", 500],
  DoIterationsD[70 000, 60 000, "Dirichlet 70K-60K", 500],
  DoIterationsD[60 000, 50 000, "Dirichlet 60K-50K", 500],
  DoIterationsD[50 000, 40 000, "Dirichlet 50K-40K", 500],
  DoIterationsD[40 000, 30 000, "Dirichlet 40K-30K", 500],
  DoIterationsD[30 000, 20 000, "Dirichlet 30K-20K", 500],
  DoIterationsD[20 000, 10 000, "Dirichlet 20K-10K", 500],
  DoIterationsD[10 000, Λ0, "Dirichlet 10K-2", 500],
  DoIterationsN[Λ0, 10 000, "Neumann 2-10K", 500],
  DoIterationsN[10 000, 20 000, "Neumann 10K-20K", 500],
  DoIterationsN[20 000, 30 000, "Neumann 20K-30K", 500],
  DoIterationsN[30 000, 40 000, "Neumann 30K-40K", 500],
  DoIterationsN[40 000, 50 000, "Neumann 40K-50K", 500],
  DoIterationsN[50 000, 60 000, "Neumann 50K-60K", 500],
  DoIterationsN[60 000, 70 000, "Neumann 60K-70K", 500],
  DoIterationsN[70 000, 80 000, "Neumann 70K-80K", 500],
  DoIterationsN[80 000, Λ1, "Neumann 80K-84K", 500]
 }]
```